import library.*;            // Access the various libraries

COMPILER Calc  $CN           // $C - compiler $N - name tokens

CHARACTERS

  digit      = "0123456789" .

  hexdigit   = digit + "ABCDEF" .

TOKENS

  decNumber  = digit { digit } .

  hexNumber  = "$" hexdigit { hexdigit } .

IGNORE CHR(0) .. CHR(31)

PRODUCTIONS

  Calc                       (. int eVal; .)

  = { Expression<out eVal>

      "="                    (. IO.writeLine(eVal); .)

    } EOF .

  Expression<out int eVal>   (. int tVal; .)

  = Term<out eVal>

    {   "+" Term<out tVal>   (. eVal = eVal + tVal; .)

      | "-" Term<out tVal>   (. eVal = eVal - tVal; .)

    } .

  Term<out int tVal>         (. int fVal; .)

  = Factor<out tVal>

    {   "*" Factor<out fVal> (. tVal = tVal * fVal; .)

      | "/" Factor<out fVal> (. if (fVal == 0)

                                  SemError("Division by zero");

                                else tVal = tVal / fVal; .)

    } .

  Factor<out int fVal>       (. fVal = 0; String num; .)

  =   decNumber              (. num = token.val;

                                try {

                                  fVal = Integer.parseInt(num);

                                } catch (NumberFormatException e) {

                                  fVal = 0;

                                  SemError("number out of range");

                                } .)

    | hexNumber              (. num = token.val.substring(1);

                                try {

                                  fVal = Integer.parseInt(num, 16);

                                } catch (NumberFormatException e) {

                                  fVal = 0;

                                  SemError("number out of range");

                                } .)

    | "(" Expression<out fVal>

      ")" .

END Calc.
